
"1. BOUNDARY CONDITIONS" 
 "THE MASS FLOW, INPUT TEMPERATURE AND SPECIFIC HEAT OF THE HEAT SOURCE 
AND  THE WATER USED FOR REFRIGERATING ARE DEFINED." 
"1.1. CICLO DE ISOPENTANO" 
 
T[10]=20 {ºC} 
m_CW=311 {kg/s} 
cp_CW=4,186 {kJ/kg*K} 
 
T[13]=195 {ºC} 
m_AR=75 {kg/s} 
cp_AR=2,34 {kJ/kg*K} 
 
 
"2. MODEL PARAMETERS" 
 
"2.1.THE WORK FLUID USED WILL BE THE ISOPENTANE" 
 
 m_cycle=37,2 
  
"2.2. Turbine" 
 
eta_isentropic_TURBINE=0,8 "Isentropic efficiency of the turbine" 
  
"2.3. Pump" 
  
eta_isentropic_PUMP=0,73 "Isentropic efficiency of the pump" 
 
"2.4. Regenerator efficiency."                    
 
eta_REGENERATOR=0,8 {-} 
 
"2.4. Heat exchanger. Condenser." 
  
A_CONDENSER_TOTAL=1213 {m^2}                  "THE TOTAL AREA OF 
THE CONDENSER IS DEFINED AS THE SUM OF THE CONDENSER (PHASE CHANGE) AND THE 
PRECONDENSER" 
A_CONDENSER_TOTAL=A_CONDENSER+A_PRECOOLER                    {m^2} 
U_CONDENSER=380 
U_PRECOOLER=392 
 
"2.5. Heat exchanger. Preheater" 

A_PREHEATER=1000 {m^2}  
U_PREHEATER=200 {W/m^2*K} 
 
"2.6.  Heat exchanger.Evaporator. Phase change." 
 
A_HEATER_TOTAL=766 {m^2}                      "THE TOTAL 
EVAPORATOR AREA IS DEFINED AS THE SUM OF THE EVAPORATOR AREA (PHASE CHANGE) 
AND OVERHEAT AREA" 
A_HEATER_TOTAL=A_HEATER+A_OVERHEATER {m^2} 
U_HEATER=305 {W/m^2*K} 
 
"2.7. Heat exchanger. Evaporator. Overheat" 
 
U_OVERHEATER=305 {W/m^2*K} 
 
 
"3. DEFINITION OF THE CYCLE" 
  
"STATE 1. HEAT RECOVERY OUTPUT AND TURBINE INPUT" 

P[1]=P[9] 
{T[1] is achieved by defining the superheater} 
  
"STATE 2. TURBINE OUTPUT AND REGENERATOR INPUT" 
  
P[2]=P[3] {bar} 
s_s[2]=s[1] {kJ/kg*K}             "ASSUMING THAT THE 
EXPANSION IS ISENTROPIC" 
  
"STATE 3. REGENERATOR OUTPUT AND PRECOOLER INPUT" 
 
P[3]=P[4] {bar} 
"h[3] DEFINED BY THE REGENERATOR'S EQUATIONS" 

 

"STATE 4. PRECOOLER OUTPUT AND CONDENSER INPUT" 

 

P[4]=P[5] {bar} 
x[4]=1        {-} 

"STATE 5. CONDENSER OUTPUT AND PUMP INPUT" 
 
P[5]=Pressure(Isopentane;T=T[5];x=x[5]) {bar} 
x[5]=0 {-} 

 

"STATE 6. PUMP OUTPUT AND REGENERATOR INPUT" 
 
P[6]=P[7] {bar} 
s_s[6]=s[5]  {kJ/kg*K}     "ASSUMING THAT THE 
EXPANSION IS ISENTROPIC" 
 
"STATE 7. REGENERATOR OUTPUT AND PREHEATER INPUT" 
 
P[7]=P[8] {bar} 



 

90 
 

"h[7] DEFINED BY THE REGENERATOR'S EQUATIONS" 
 
"STATE 8. PREHEATER OUTPUT AND HEATER INPUT" 
 
P[8]=P[9] {bar} 
x[8]=0 {-} 
 
"STATE 9. HEATER OUTPUT AND OVERHEATER INPUT" 
 
P[9]=Pressure(Isopentane;T=T[9];x=x[9]) {bar} 
x[9]=1 {-} 
 
"4. THERMODYNAMIC PROPERTIES" 
  
"STATE 1. EXIT OF THE HEAT RECOVERY AND TURBINE INLET" 
  
h[1]=Enthalpy(Isopentane;T=T[1];P=P[1]) {kJ/kg} 
s[1]=Entropy(Isopentane;T=T[1];P=P[1]) {kJ/kg*K} 
  
"STATE 2. TURBINE OUTPUT AND REGENERATOR INPUT" 
 
h_s[2]=Enthalpy(Isopentane;P=P[2];s=s_s[2]) {kJ/kg} 
eta_isentropic_TURBINE=(h[1]-h[2])/(h[1]-h_s[2]) {-} 
T[2]=Temperature(Isopentane;P=P[2];h=h[2]) {ºC} 
s[2]=Entropy(Isopentane;P=P[2];h=h[2]) {kJ/kg*K} 
 
"STATE 3. REGENERATOR OUTPUT AND PRECOOLER INPUT" 
 
T[3]=Temperature(Isopentane;P=P[3];h=h[3]) {ºC} 
s[3]=Entropy(Isopentane;P=P[3];h=h[3])    {kJ/kg*K} 

"STATE 4. PRECOOLER OUTPUT AND CONDENSER INPUT" 
T[4]=Temperature(Isopentane;P=P[4];x=x[4]) {ºC} 
h[4]=Enthalpy(Isopentane;P=P[4];x=x[4]) {kJ/kg} 
s[4]=Entropy(Isopentane;P=P[4];x=x[4])         {kJ/kg*K} 

"STATE 5. CONDENSER OUTPUT AND PUMP INPUT" 
 
h[5]=Enthalpy(Isopentane;P=P[5];x=x[5]) {kJ/kg} 
s[5]=Entropy(Isopentane;P=P[5];x=x[5]) {kJ/kg*K} 
 
"STATE 6. PUMP OUTPUT AND REGENERATOR INPUT" 
 
h_s[6]=Enthalpy(Isopentane;P=P[6];s=s_s[6]) {kJ/kg} 
eta_isentropic_PUMP=(h_s[6]-h[5])/(h[6]-h[5]) {-} 
T[6]=Temperature(Isopentane;P=P[6];h=h[6]) {ºC} 
s[6]=Entropy(Isopentane;P=P[6];h=h[6]) {kJ/kg*K} 
 
"STATE 7. REGENERATOR OUTPUT AND PREHEATER INPUT" 
 
T[7]=Temperature(Isopentane;P=P[7];h=h[7]) {ºC} 
s[7]=Entropy(Isopentane;P=P[7];h=h[7]) {kJ/kg*K} 
 
"STATE 8. PREHEATER OUTPUT AND HEATER INPUT" 
 
T[8]=Temperature(Isopentane;P=P[8];x=x[8]) {ºC} 

h[8]=Enthalpy(Isopentane;P=P[8];x=x[8]) {kJ/kg} 
s[8]=Entropy(Isopentane;P=P[8];x=x[8]) {kJ/kg*K} 
 
"STATE 9. HEATER OUTPUT AND OVERHEATER INPUT" 
 
h[9]=Enthalpy(Isopentane;P=P[9];x=x[9]) {kJ/kg} 
s[9]=Entropy(Isopentane;P=P[9];x=x[9]) {kJ/kg*K} 
 
"5. COMPONENTS" 
  
"TURBINE, STATE 1 TO 2" 
  
W_TURBINE=m_cycle*(h[1]-h[2]) {kW} 
 
"PUMP, STATE 5 TO 6" 
 
W_PUMP=m_cycle*(h[6]-h[5]) {kW} 
 
"REGENERATOR, STATE 2 TO 3 AND 6 TO 7" 
 
cp_media_prim=abs((h[2]-h[3])/(T[2]-T[3]))                                             {kJ/kg*K} 
cp_media_sec=abs((h[7]-h[6])/(T[7]-T[6]))                                             {kJ/kg*K} 
cp_min=min(cp_media_prim;cp_media_sec)                                             {kJ/kg*K} 
  
Q_MAX_REGENERADOR=cp_min*m_cycle*(T[2]-T[6])  {kW} 
Q_REGENERADOR=Q_MAX_REGENERADOR*eta_REGENERADOR {kW} 
 
Q_REGENERADOR=m_cycle*(h[2]-h[3])  {kW} 
Q_REGENERADOR=m_cycle*(h[7]-h[6])  {kW} 
 
"CONDENSER" 
"PRE-COOLER, FROM STATE 3 TO 4" 
 
Q_PRECONDENSACION=m_cycle*(h[3]-h[4])  {kW} 
Q_PRECONDENSACION=m_CW*cp_CW*(T[12]-T[11])  {kW} 
Q_PRECONDENSACION*1000=UA_PRECONDENSACION*LMTD_PRECONDENSACION {kW}  

  
LMTD_PRECONDENSACION=((T[4]-T[11])-(T[3]-T[12]))/ln((T[4]-T[11])/(T[3]-T[12])) {ºC} 
  
UA_PRECONDENSACION=U_PRECONDENSACION*A_PRECONDENSACION {W/K} 
 
"PHASE CHANGE FROM STATE 3 TO 4"  
 
Q_COND=m_cycle*(h[4]-h[5])  {kW} 
Q_COND=m_CW*cp_CW*(T[11]-T[10])  {kW} 
Q_COND*1000=UA_COND*LMTD_COND  {kW}

 
LMTD_COND=((T[5]-T[10])-(T[4]-T[11]))/ln((T[5]-T[10])/(T[4]-T[11])) {ºC} 
  
UA_COND=U_COND*A_COND  {W/K} 
 
"HEAT RECOVERY" 
  
"PREHEATER, STATE 7 TO 8" 

Q_PREHEATER=m_cycle*(h[8]-h[7])  {kW} 
Q_PREHEATER=m_AR*cp_AR*(T[15]-T[16])  {kW}  
Q_PREHEATER*1000=UA_PREHEATER*LMTD_PREHEATER        {kW} 
 
 
LMTD_PREHEATER=((T[16]-T[7])-(T[15]-T[8]))/ln((T[16]-T[7])/(T[15]-T[8])) {ºC} 
  
UA_PREHEATER=U_PREHEATER*A_PREHEATER  {W/K} 
 
"EVAPORATOR, FROM STATE 8 TO 9" 
 
Q_HEATER=m_cycle*(h[9]-h[8])  {kW} 
Q_HEATER=m_AR*cp_AR*(T[14]-T[15])  {kW} 
Q_HEATER*1000=UA_HEATER*LMTD_HEATER  {kW} 

 
LMTD_HEATER=((T[15]-T[8])-(T[14]-T[9]))/ln((T[15]-T[8])/(T[14]-T[9])) {ºC} 
  
UA_HEATER=U_HEATER*A_HEATER  {W/K} 
 
"OVERHEATING, STATE 9 TO 1" 
 
Q_OVERHEATER=m_cycle*(h[1]-h[9])  {kW} 
Q_OVERHEATER*1000=UA_OVERHEATER*LMTD_OVERHEATER         {kW} 
Q_OVERHEATER=m_AR*cp_AR*(T[13]-T[14])  {kW} 

 
LMTD_OVERHEATER=((T[14]-T[9])-(T[13]-T[1]))/ln((T[14]-T[9])/(T[13]-T[1])) {ºC} 
  
UA_OVERHEATER=U_OVERHEATER*A_OVERHEATER {W/K} 
 
"6. RESULTS" 
W_total=W_TURBINE-W_PUMP {kW} 
eta_cycle=(W_TURBINE-W_PUMP)/Q_IN {-} 
 
Q_IN=m_cycle*(h[1]-h[7]) {kW} 
Q_OUT=m_cycle*(h[3]-h[5]) {kW} 
 
 
 
 
 
 
 
 
 

 

 

 

 


